

Язык запросов 1С:Предприятия 8 позволяет использовать временные таблицы в запросах. Использование временных таблиц помогает повысить производительность запросов и сделать текст сложных запросов более легким для восприятия.

Работа с временными таблицами обеспечивается двумя составляющими:

объектом встроенного языка **МенеджерВременныхТаблиц**, который хранит в себе данные временных таблиц;

синтаксисом языка запросов, позволяющим создавать новые временные таблицы и использовать существующие временные таблицы.

Менеджер временных таблиц

Менеджер временных таблиц предназначен для управления временем жизни временных таблиц, создаваемых в процессе работы прикладного решения.

В одном прикладном решении может быть создано произвольное количество экземпляров менеджера временных таблиц, каждый из которых хранит свой набор временных таблиц. Каждая временная таблица однозначно идентифицируется своим именем, и в пределах одного менеджера временных таблиц все временные таблицы должны иметь уникальные имена. Экземпляр менеджера временных таблиц может быть создан с помощью конструктора Новый.

Например:

МенеджерВременныхТаблиц = Новый МенеджерВременныхТаблиц;

Все временные таблицы, созданные в данном экземпляре менеджера, существуют до тех пор, пока существует сам экземпляр менеджера временных таблиц. При уничтожении экземпляра менеджера все временные таблицы, содержащиеся в нем, также удаляются. Менеджер временных таблиц можно закрыть принудительно при помощи метода **Закрыть()**. При этом будут удалены все созданные в нем таблицы. Дальнейшая работа с данным экземпляром менеджера будет невозможна.

Создание временных таблиц

Создание временных таблиц осуществляется с помощью объекта Запрос встроенного языка 1С:Предприятия 8. Связь запроса с менеджером временных таблиц осуществляется с помощью свойства МенеджерВременныхТаблиц запроса, в котором указывается тот экземпляр менеджера, в котором должны создаваться временные таблицы.

Например:

МенеджерВременныхТаблиц = Новый МенеджерВременныхТаблиц;

Запрос = Новый Запрос;

Запрос.МенеджерВременныхТаблиц = МенеджерВременныхТаблиц;

Временная таблица может быть создана на основе данных базы данных или на основе внешнего источника данных (например, таблицы значений). Для того чтобы создать временную таблицу на основе данных базы данных, следует установить объекту Запрос менеджер временных таблиц, а затем выполнить запрос к базе данных, используя ключевое слово **ПОМЕСТИТЬ**, после которого указать имя создаваемой временной таблицы. Ключевое слово ПОМЕСТИТЬ располагается после списка выборки запроса.

Например:

ВЫБРАТЬ

Код,

Наименование

ПОМЕСТИТЬ ВременнаяТаблица

ИЗ Справочник.Номенклатура

Результат исполнения такого запроса будет содержать одну строку с одной колонкой с именем "Количество", в которой будет находиться количество записей, помещенных в созданную таблицу. В случае если менеджер временных таблиц не установлен, или менеджер был закрыт, будет выдана ошибка. В случае если в установленном менеджере временных таблиц уже существует таблица с указанным именем, будет выдана ошибка. При необходимости создания индекса для временной таблицы, следует в запросе указать ключевое слово ИНДЕКСИРОВАТЬ ПО, после которого перечислить поля, по которым нужно построить индекс.

Например:

ВЫБРАТЬ

Код,

Наименование

ПОМЕСТИТЬ ВременнаяТаблица

ИЗ Справочник.Номенклатура

ИНДЕКСИРОВАТЬ ПО Код

Поля, по которым происходит индексирование должны находиться в списке выборки.

Для того, чтобы создать временную таблицу и заблокировать данные таблиц, на основании которых создается временная, следует использовать конструкцию **ДЛЯ ИЗМЕНЕНИЯ**, следует использовать конструкцию **ДЛЯ ИЗМЕНЕНИЯ**.

Например:

ВЫБРАТЬ

РасходнаяНакладная.Ссылка,

РасходнаяНакладная.Номер,

РасходнаяНакладная.Дата

ПОМЕСТИТЬ ВременнаяТаблица

ИЗ

Документ.РасходнаяНакладная КАК РасходнаяНакладная

ГДЕ

РасходнаяНакладная.Ссылка В(&Документы)

ДЛЯ ИЗМЕНЕНИЯ

Для того чтобы создать временную таблицу на основании внешнего источника, следует в тексте запроса в списке источников указать имя параметра, в который будет помещен внешний источник. Остальной синтаксис идентичен обычному созданию временной таблицы. В качестве внешнего источника могут выступать:

таблица значений;

табличная часть;

результат запроса.

Ниже приведен пример создания временной таблицы на основе внешнего источника:

ВЫБРАТЬ

Код,

Наименование

ПОМЕСТИТЬ ВременнаяТаблица

ИЗ &ВнешнийИсточник КАК ВнешнийИсточник

В данном примере во временную таблицу "ВременнаяТаблица" будет помещено содержимое колонок "Код" и "Наименование" из внешнего источника, например, таблицы значений, переданной в качестве параметра "ВнешнийИсточник". В случае если временная таблица создается на основании внешнего источника, в запросе нельзя использовать объединения и соединения. Также нельзя использовать поля, являющиеся реквизитами элементов таблиц, на основании которых создается временная таблица.

Использование временных таблиц

Для использования существующих временных таблиц в запросе следует установить объекту Запрос менеджер временных таблиц, после чего к временным таблицам, содержащимся в данном менеджере временных таблиц, можно обращаться по имени, как к обычным таблицам запроса.

Удаление временных таблиц

Для удаления временной таблицы из менеджера временных таблиц следует воспользоваться ключевым словом языка запроса УНИЧТОЖИТЬ, после которого указывается имя уничтожаемой таблицы. Например:

УНИЧТОЖИТЬ ВременнаяТаблица

В случае если уничтожаемой таблицы не существует, будет выдана ошибка.

Пример

Цитата

Процедура ОбработкаПроведения(Отказ, РежимПроведения)

МВТ = Новый МенеджерВременныхТаблиц;

Запрос = Новый Запрос;

Запрос.МенеджерВременныхТаблиц = МВТ;

Запрос.Текст = "

| ВЫБРАТЬ

| Номенклатура, СУММА(Количество) КАК Количество

| ПОМЕСТИТЬ ДокТЧ

| ИЗ

| Документ.Расходная.Товары

| ГДЕ

| Ссылка = &Ссылка

| СГРУППИРОВАТЬ ПО Номенклатура";

Запрос.УстановитьПараметр("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить(); //Прим. 1

Запрос = Новый Запрос;

Запрос.МенеджерВременныхТаблиц = МВТ;

Запрос.Текст = "ВЫБРАТЬ РАЗЛИЧНЫЕ

| Номенклатура

| ПОМЕСТИТЬ СписокТоваров

| ИЗ

| Документ.Расходная.Товары

| ГДЕ

| Ссылка = &Ссылка";

Запрос.УстановитьПараметр("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить(); //Прим. 2

Запрос = Новый Запрос;

Запрос.МенеджерВременныхТаблиц = МВТ;

Запрос.Текст = "

| ВЫБРАТЬ

| Док.Номенклатура,

| Док.Количество КАК Док_Количество,

| ЕСТЬNULL(Рег.КоличествоОстаток,0) КАК Рег_Количество

| ИЗ

| ДокТЧ КАК Док

| ЛЕВОЕ СОЕДИНЕНИЕ

| РегистрНакопления.ОстаткиТоваров.Остатки(,

| Номенклатура В(ВЫБРАТЬ РАЗЛИЧНЫЕ

| Номенклатура

| ИЗ

| СписокТоваров КАК СписокТоваров)) КАК Рег

| ПО

| Док.Номенклатура = Рег.Номенклатура";

РезультатЗапроса = Запрос.Выполнить();

Выборка = РезультатЗапроса.Выбрать();

Пока Выборка.Следующий() Цикл

//Проверка отрицательных остатков

//Проведение по регистру

КонецЦикла;

КонецПроцедуры

Используйте временные таблицы осторожно, т.к. в зависимости от их применения они могут храниться временно и на жестком диске, что замедлит скорость выполнения.